

# An Algorithm for Computing Coefficients of Words in Expressions Involving Exponentials and its Application to the Construction of Exponential Integrators

Harald Hofstätter<sup>1</sup>[0000-0003-0638-9611], Winfried Auzinger<sup>2</sup>[0000-0002-9631-2601], and Othmar Koch<sup>3</sup>[0000-0002-1956-221X]

<sup>1</sup> Reitschachersiedlung 4/6, A-7100 Neusiedl am See, Austria  
hofi@harald-hofstaetter.at <http://harald-hofstaetter.at>

<sup>2</sup> Vienna University of Technology, Institute of Analysis and Scientific Computing,  
Wiedner Hauptstraße 8-10, A-1040 Wien, Austria w.auzinger@tuwien.ac.at  
<http://asc.tuwien.ac.at/~winfried>

<sup>3</sup> University of Vienna, Institute of Mathematics, Oskar-Morgenstern-Platz 1, A-1090  
Wien, Austria othmar@othmar-koch.org <http://www.othmar-koch.org>

**Abstract.** This paper discusses an efficient implementation of the generation of order conditions for the construction of exponential integrators like exponential splitting and Magnus-type methods in the computer algebra system Maple. At the core of this implementation is a new algorithm for the computation of coefficients of words in the formal expansion of the local error of the integrator. The underlying theoretical background including an analysis of the structure of the local error is briefly reviewed. As an application the coefficients of all 8th order self-adjoint commutator-free Magnus-type integrators involving the minimum number of 8 exponentials are computed.

**Keywords:** Splitting methods · Magnus-type integrators · Local error · Order conditions · Computer algebra.

## 1 Introduction

In the construction of integration schemes for the numerical solution of evolution equations the coefficients of the schemes are usually determined as solutions of certain systems of polynomial equations. The obvious requirement that all terms up to a certain order  $O(\tau^p)$  in the Taylor expansion of the local error with respect to the step-size  $\tau$  vanish, usually leads, if applied in a naive way, to a far over-determined system of equations. In some cases, however, due to the special structure of the local error, the fact that a small subset of terms vanishes already implies that all terms up to order  $O(\tau^p)$  in the Taylor expansion vanish. This leads to a minimal, non-redundant system of equations, the so-called *order conditions*, see for instance [2].

As it turns out, this is in particular the case for (generalized) exponential splitting methods for the numerical solution of evolution equations of the form<sup>4</sup>

$$\partial_t u(t) = Au(t) + Bu(t), \quad t \geq t_0, \quad u(t_0) = u_0, \quad A, B \in \mathbb{C}^{d \times d}$$

(and also for Magnus-type integrators considered below). Here one step  $u_{n+1} = \mathcal{S}(\tau)u_n$  with step-size  $\tau$  is specified by an approximation  $\mathcal{S}(\tau) = \mathcal{S}_J(\tau) \cdots \mathcal{S}_1(\tau)$  of the exact solution operator  $\mathcal{E}(\tau) = e^{\tau(A+B)}$ , where the factors  $\mathcal{S}_j(\tau)$  are exponentials whose applications  $\mathcal{S}_j(\tau)y$  to vectors  $y \in \mathbb{C}^d$  can be effectively computed. Prototypical examples are for instance the classical second order Strang splitting

$$\mathcal{S}(\tau) = e^{\frac{1}{2}\tau B} e^{\tau A} e^{\frac{1}{2}\tau B}, \quad (1)$$

or the 4th order generalized splitting

$$\mathcal{S}(\tau) = e^{\frac{1}{6}\tau B} e^{\frac{1}{2}\tau A} e^{\frac{2}{3}\tau B + \frac{1}{72}\tau^3[B, [A, B]]} e^{\frac{1}{2}\tau A} e^{\frac{1}{6}\tau B} \quad (2)$$

proposed in [5, 13].

The analysis of the structure of the local error

$$\mathcal{L}(\tau) = \mathcal{S}(\tau) - \mathcal{E}(\tau) = \mathcal{S}_J(\tau) \cdots \mathcal{S}_1(\tau) - e^{\tau(A+B)}$$

and the resulting derivation of order conditions is advantageously carried out in a purely formal way by introducing non-commutative symbols  $A, B$  representing respectively  $\tau A, \tau B$ , and considering the formal expression corresponding to  $\mathcal{L}(\tau)$  with  $\tau A, \tau B$  substituted by  $A, B$ . Thus, with a slight generalization anticipating an application to Magnus-type integrators, we study expressions of the form

$$X = e^{\Phi_J} \cdots e^{\Phi_1} - e^{\Omega}, \quad (3)$$

where  $\Phi_1, \dots, \Phi_J$ , and  $\Omega$  are linear combinations of non-commutative symbols and commutators thereof. In particular, in Theorem 2 of Section 3 the structure of the leading term in the formal series expansion of such expressions is characterized, which leads to the derivation of order conditions in Theorem 3. These theoretical considerations of Section 3 are essentially a review of the theory developed in [9, Section 2], which extends and generalizes results from [2].

The main focus of the present paper is on a concrete implementation of the generation of order conditions according to Theorem 3 in the computer algebra system Maple.<sup>5</sup> This can be realized in a very efficient way, utilizing a new algorithm derived in [9] for the computation of coefficients of words (i.e., finite products of non-commutative symbols) in expressions like (3) involving exponentials, whose concise Maple implementation is reproduced in its entirety

<sup>4</sup> For our considerations, it is sufficient to discuss linear problems. The algebraic structure underlying method construction is the same for nonlinear problems due to the calculus of Lie derivatives [8, Section III.5.1].

<sup>5</sup> We have used Maple 18, Maple is a trademark of Waterloo Maple Inc.

in Section 2.<sup>6</sup> Our approach is thus a relevant contribution compared to previous work on the generation of order conditions, see, e.g., [8, Section III.5] or [3].

Finally in Section 4, after a brief review of material from [1, 9] on Magnus-type integrators for the numerical solution of non-autonomous evolution equations of the form

$$\partial_t u(t) = A(t)u(t), \quad t \geq t_0, \quad u(t_0) = u_0, \quad A(t) \in \mathbb{C}^{d \times d},$$

we consider a non-trivial application of the theory of Section 3: We compute in a systematic way the coefficients of all 8th order commutator-free self-adjoint Magnus-type methods involving the minimum number of 8 exponentials.

## 2 Coefficients of Words in Expressions Involving Exponentials

Let  $\mathcal{A}$  denote a fixed set of non-commutative variables. Given an expression  $X$  in these variables involving exponentials like (3) we want to calculate real or complex coefficients

$$c_w = \text{coeff}(w, X), \quad w \in \mathcal{A}^*$$

in the formal expansion

$$X = \sum_{w \in \mathcal{A}^*} c_w w \in \mathbb{C}\langle\langle\mathcal{A}\rangle\rangle.$$

$X$  is thus represented as an element of  $\mathbb{C}\langle\langle\mathcal{A}\rangle\rangle$ , the algebra of formal power series in the non-commutative variables in  $\mathcal{A}$ . Here,  $\mathcal{A}^*$  denotes the set of all words over the alphabet  $\mathcal{A}$ , i.e., the set of all finite products (including the empty product  $\text{Id}$ ) of elements of  $\mathcal{A}$ .

### 2.1 A Family of Homomorphisms

In [9] an efficient algorithm for the computation of  $\text{coeff}(w, X)$  was derived, which is based on a suitably constructed family of maps  $\{\varphi_w : w \in \mathcal{A}^*\}$ , where for each word  $w = w_1 \cdots w_{\ell(w)} \in \mathcal{A}^*$  of length  $\ell(w) \geq 1$ ,  $\varphi_w(X)$  is an upper triangular matrix in  $\mathbb{C}^{(\ell(w)+1) \times (\ell(w)+1)}$  whose entries are coefficients of subwords of  $w$  in  $X$ ,

$$\varphi_w(X)_{i,j} = \begin{cases} \text{coeff}(w_{i:j-1}, X), & \text{if } i < j, \\ \text{coeff}(\text{Id}, X), & \text{if } i = j, \\ 0, & \text{if } i > j. \end{cases} \quad (4)$$

Here  $w_{i:j-1} = w_i w_{i+1} \cdots w_{j-1}$  denotes the subword of  $w$  of length  $j - i$ , starting at position  $i$  and ending at position  $j - 1$ .

---

<sup>6</sup> All Maple code discussed in this paper is also provided by the package **Expocon** available at [7]. Additionally, this package includes routines for the generation of Lyndon words and Lyndon bases. For simplicity, such words and bases have always been hardcoded whenever needed in the code examples of this paper.

**Theorem 1 ([9, Theorem 2.4]).** *The map  $\varphi_w$  defined by (4) is an algebra homomorphism*

$$\mathbb{C}\langle\langle\mathcal{A}\rangle\rangle \rightarrow \mathbb{C}^{(\ell(w)+1) \times (\ell(w)+1)},$$

i.e.,

(i)  $\varphi_w$  is linear,

$$\varphi_w(\alpha X + \beta Y) = \alpha \varphi_w(X) + \beta \varphi_w(Y), \quad X, Y \in \mathbb{C}\langle\langle\mathcal{A}\rangle\rangle, \quad \alpha, \beta \in \mathbb{C};$$

(ii)  $\varphi_w$  preserves the multiplicative structure,

$$\varphi_w(X \cdot Y) = \varphi_w(X) \cdot \varphi_w(Y), \quad X, Y \in \mathbb{C}\langle\langle\mathcal{A}\rangle\rangle.$$

Furthermore, if  $\text{coeff}(\text{Id}, X) = 0$ , then

$$\varphi_w(\exp X) = \exp \varphi_w(X),$$

where the exponential of the strictly upper triangular and thus nilpotent matrix  $\varphi_w(X)$  is exactly computable in a finite number of steps.

## 2.2 Maple Implementation of the Algorithm

It follows that for a given expression  $X$ , a recursive application of  $\varphi_w$  (the recursion terminates with well-defined values  $\varphi_w(a)$  for the “atoms”  $a \in \mathcal{A}$ ) yields  $\varphi_w(X)$ , from which one can read off  $\text{coeff}(w, X)$  as the element at the upper right corner,

$$\text{coeff}(w, X) = \varphi_w(X)_{1, \ell(w)+1},$$

cf. (4). By organizing this calculation in a more efficient way, the function **phiv** defined in the Maple code displayed below computes

$$\mathbf{phiv}(w, X, v) = \varphi_w(X) \cdot v$$

for a vector  $v \in \mathbb{C}^{\ell(w)+1}$  without explicitly generating the matrix  $\varphi_w(X)$ . It recursively traverses the expression tree representing the expression  $X$ . At each node of the tree the evaluation branches out depending on whether the current node represents

- a non-commutative symbol (the atomic case which terminates the recursion),
- a sum of subexpressions,
- a product of subexpressions,
- a power of a subexpression,
- a commutator of subexpressions, or
- an exponential of a subexpression.

Finally, the function **wcoeff**<sup>7</sup> computes  $\text{coeff}(w, X)$  via

$$\text{coeff}(w, X) = \text{first component of } \mathbf{phiv}(w, X, (0, \dots, 0, 1)^T).$$

<sup>7</sup> The function was called **wcoeff** because **coeff** is already defined in Maple.

The elements of the alphabet  $\mathcal{A}$  are represented within Maple as non-commutative symbols, which are provided by the package `Physics`. Note that except for providing such non-commutative symbols (and the type `Commutator`) we do not need or use any further feature of the package `Physics`. Words  $w \in \mathcal{A}^*$  are represented as lists of non-commutative symbols.

```
> with(Physics):
> phiv := proc (w, X, v)
    local i, v1, v2, f, zero;
    if type(X, name) and type(X, noncommutative) then
        return [seq('if'(op(i,w)=X, v[i+1], 0), i=1..nops(w)), 0]
    elif type(X, '+') then
        return add(phiv(w, op(i, X), v), i=1..nops(X))
    elif type(X, '*') then
        v1 := v; zero := [0$nops(w)+1];
        for i from nops(X) to 1 by -1 do
            v1 := phiv(w, op(i, X), v1);
            if v1=zero then return zero end if;
        end do;
        return v1
    elif type(X, anything^integer) then
        v1 := v; zero := [0$nops(w)+1];
        for i from 1 to op(2, X) do
            v1 := phiv(w, op(1, X), v1);
            if v1=zero then return zero end if;
        end do;
        return v1
    elif (type(X, function) and
        op(0, X) = Physics[Commutator]) then
        return phiv(w, op(1, X), phiv(w, op(2, X), v))
            - phiv(w, op(2, X), phiv(w, op(1, X), v))
    elif type(X, exp(anything)) then
        v1 := v; v2 := v; zero := [0$nops(w)+1]; f := 1;
        for i from 1 to nops(w) do
            f := f*i; v1 := phiv(w, op(X), v1);
            if v1=zero then return v2 end if;
            v2 := v2 + v1/f;
        end do;
        return v2
    end if;
    return [seq(X*x, x=v)]
end proc:
> wcoeff := proc (w, X)
    return phiv(w, X, [0$nops(w), 1])[1]
end proc:
```

### 3 Order Conditions for Exponential Integrators

In this section we review the theory developed in [9, Section 2], which extends and generalizes results from [2]. We consider expressions of the form (3),

$$X = e^{\Phi_J} \dots e^{\Phi_1} - e^{\Omega}, \quad (5)$$

where the exponents  $\Phi_1, \dots, \Phi_J$ , and  $\Omega$  are linear combinations of non-commutative symbols and commutators thereof, i.e., elements of  $[\mathbb{C}\langle\mathcal{A}\rangle]$ , the free Lie algebra generated by the non-commutative symbols of a given alphabet  $\mathcal{A}$ , which in a natural way is embedded in the algebra  $\mathbb{C}\langle\langle\mathcal{A}\rangle\rangle$  of formal power series in these symbols.

In the applications we are interested in,  $S = e^{\Phi_J} \dots e^{\Phi_1}$  represents an exponential integrator for the numerical solution of an evolution equation, and  $E = e^{\Omega}$  represents the exact local solution operator for this equation. We can interpret (5) as the error of the approximation  $S$  of  $E$ , i.e.,  $X$  represents the *local error* of the exponential integrator  $S$ .

#### 3.1 Grading of Words and Homogeneous Lie Elements

We consider a grading function on the alphabet  $\mathcal{A}$ ,

$$\text{grade}(a) \in \{1, 2, \dots\}, \quad a \in \mathcal{A}, \quad (6)$$

and extend it to words  $w = w_1 \dots w_{\ell(w)} \in \mathcal{A}^*$  by

$$\text{grade}(w) = \sum_{j=1}^{\ell(w)} \text{grade}(w_j).$$

We call  $\Psi \in [\mathbb{C}\langle\mathcal{A}\rangle]$  a homogeneous Lie element of grade  $q$  if it can be expanded in  $\mathbb{C}\langle\langle\mathcal{A}\rangle\rangle$  to a linear combination of words all of the same grade  $q$ . The decomposition

$$[\mathbb{C}\langle\mathcal{A}\rangle] = \bigoplus_{q=1}^{\infty} \mathfrak{g}_q, \quad \mathfrak{g}_q = \{\text{homogeneous Lie elements of grade } q\} \quad (7)$$

into a direct sum of subspaces makes  $[\mathbb{C}\langle\mathcal{A}\rangle]$  a graded Lie algebra, cf. [12].

*Remark 1.* In the applications we are interested in, the symbols  $a \in \mathcal{A}$  represent objects which depend on a (small) parameter  $\tau > 0$  (for instance, a time increment). The grading (6) is chosen such that it reflects the order of magnitude of the represented objects,

$$a \simeq O(\tau^{\text{grade}(a)}), \quad a \in \mathcal{A}.$$

For example, in the case of an application to splitting methods with step-size  $\tau$ ,

$$\mathcal{A} = \{\mathbf{A}, \mathbf{B}\}, \quad \mathbf{A} \simeq \tau \mathbf{A} = O(\tau), \quad \mathbf{B} \simeq \tau \mathbf{B} = O(\tau) \Rightarrow \text{grade}(\mathbf{A}) = \text{grade}(\mathbf{B}) = 1,$$

cf. Section 1.

### 3.2 Leading Error Term

The following theorem states that the leading error term  $\Theta$  of the approximation  $e^{\Phi_J} \dots e^{\Phi_1}$  of  $e^\Omega$  is a homogeneous Lie element of some grade  $q$ .

**Theorem 2 ([9, Theorem 2.1]).** *If for  $\Phi_1, \dots, \Phi_J, \Omega \in [\mathbb{C}\langle\mathcal{A}\rangle]$  the expression  $e^{\Phi_J} \dots e^{\Phi_1} - e^\Omega$  is expanded in  $\mathbb{C}\langle\mathcal{A}\rangle$  as*

$$e^{\Phi_J} \dots e^{\Phi_1} - e^\Omega = \sum_{w \in \mathcal{A}^*} c_w w = \Theta + R,$$

where

$$\Theta = \sum_{\text{grade}(w)=q_{\min}} c_w w, \quad q_{\min} = \min\{\text{grade}(w) : w \in \mathcal{A}^*, c_w \neq 0\} \quad (8)$$

(and the remainder  $R$  contains the terms of grade  $> q_{\min}$ ), then  $\Theta$  can be represented as a linear combination of commutators, i.e.,  $\Theta$  is a homogeneous Lie element of grade  $q_{\min}$ .

To illustrate Theorem 2 we consider as an example

$$X = e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B} - e^{A+B}, \quad \mathcal{A} = \{A, B\}, \quad \text{grade}(A) = \text{grade}(B) = 1,$$

cf. (1). The following Maple code computes the coefficients of all words of length  $\leq 3$  (i.e., of all words  $w$  with  $\text{grade}(w) \leq 3$ ) in  $X$ .

```
> Physics[Setup](noncommutativeprefix = {A, B}):
> X := exp((1/2)*B)*exp(A)*exp((1/2)*B)-exp(A+B):
> W := [[A], [B], [A, A], [A, B], [B, A], [B, B],
        [A, A, A], [A, A, B], [A, B, A], [A, B, B],
        [B, A, A], [B, A, B], [B, B, A], [B, B, B]]:
> seq(wcoeff(w, X), w in W);
```

$$0, 0, 0, 0, 0, 0, 0, \frac{1}{12}, \frac{-1}{6}, \frac{-1}{24}, \frac{1}{12}, \frac{1}{12}, \frac{-1}{24}, 0$$

It follows

$$\begin{aligned} e^{\frac{1}{2}B} e^A e^{\frac{1}{2}B} - e^{A+B} &= \frac{1}{12} AAB - \frac{1}{6} ABA - \frac{1}{24} ABB + \frac{1}{12} BAA + \frac{1}{12} BAB - \frac{1}{24} BBA + \dots \\ &= \frac{1}{12} [A, [A, B]] - \frac{1}{24} [[A, B], B] + \dots \end{aligned}$$

Here the leading error term is indeed a homogeneous Lie element of grade 3.

### 3.3 Symmetry

A product of exponentials  $S = e^{\Phi_J} \dots e^{\Phi_1}$ ,  $\Phi_j \in [\mathbb{C}\langle\mathcal{A}\rangle]$  is called self-adjoint or symmetric,<sup>8</sup> if

$$\Phi_{J-j+1} = \sum_k (-1)^{k+1} X_{j,k}, \quad j = 1, \dots, J$$

<sup>8</sup> For an exponential integrator represented by  $S$  this definition conforms with the usual definition of a self-adjoint integrator, e.g.,  $\mathcal{S}(-\tau)\mathcal{S}(\tau) = \text{Id}$  for a generalized splitting method where  $\mathcal{S}(\tau)$  is  $S$  with  $\tau A, \tau B$  substituted for  $A, B$ .

**Table 1.** Lyndon words  $\mathcal{W}_q$  of grade  $q$  and Lyndon basis  $\mathcal{B}_q$  of  $\mathfrak{g}_q$  for  $\mathcal{A} = \{\mathbf{A}, \mathbf{B}\}$  and  $\text{grade}(\mathbf{A}) = \text{grade}(\mathbf{B}) = 1$ .

$q$	Lyndon words	Lyndon basis
1	$\mathbf{A}, \mathbf{B}$	$\mathbf{A}, \mathbf{B}$
2	$\mathbf{AB}$	$[\mathbf{A}, \mathbf{B}]$
3	$\mathbf{AAB}, \mathbf{ABB}$	$[\mathbf{A}, [\mathbf{A}, \mathbf{B}]], [[\mathbf{A}, \mathbf{B}], \mathbf{B}]$
4	$\mathbf{AAAB}, \mathbf{AABB}, \mathbf{ABBB}$	$[\mathbf{A}, [\mathbf{A}, [\mathbf{A}, \mathbf{B}]]], [\mathbf{A}, [[\mathbf{A}, \mathbf{B}], \mathbf{B}]], [[[\mathbf{A}, \mathbf{B}], \mathbf{B}], \mathbf{B}]$
5	$\mathbf{AAAA}, \mathbf{AAAB}, \mathbf{AABAB}, \mathbf{AABBB}, \mathbf{ABABB}, \mathbf{ABBBB}$	$[\mathbf{A}, [\mathbf{A}, [\mathbf{A}, [\mathbf{A}, \mathbf{B}]]]], [\mathbf{A}, [\mathbf{A}, [[\mathbf{A}, \mathbf{B}], \mathbf{B}]]], [[[\mathbf{A}, [\mathbf{A}, \mathbf{B}]], [\mathbf{A}, \mathbf{B}]]], [\mathbf{A}, [[[\mathbf{A}, \mathbf{B}], \mathbf{B}], \mathbf{B}]], [[[\mathbf{A}, \mathbf{B}], [[\mathbf{A}, \mathbf{B}], \mathbf{B}]]], [[[[\mathbf{A}, \mathbf{B}], \mathbf{B}], \mathbf{B}], \mathbf{B}]$

holds, where the  $X_{j,k} \in \mathfrak{g}_k$  are the components of  $\Phi_j = \sum_k X_{j,k}$  with respect to the decomposition (7). It follows that a single exponential  $e^\Phi$  is self-adjoint, if and only if  $\Phi$  is a sum of homogeneous Lie elements of odd grade,  $\Phi = X_1 + X_3 + \dots$ ,  $X_k \in \mathfrak{g}_k$ . It was proved in [9, Theorem 2.2] that in (8) the grade  $q_{\min}$  of the homogeneous Lie element  $\Theta$  is necessarily odd if  $e^{\Phi_J} \dots e^{\Phi_1}$  and  $e^\Omega$  are both self-adjoint.

### 3.4 Lyndon words and Lyndon bases

For a homogeneous Lie element  $\Theta$  of grade  $q$  like the one given in (8) let

$$\Theta = \sum_{b \in \mathcal{B}_q} c_b b \quad (9)$$

be its representation in a basis  $\mathcal{B}_q$  of the subspace  $\mathfrak{g}_q$  of (7). Furthermore, let  $\mathcal{W}_q \subset \mathcal{A}^*$  be a set of words of grade  $q$  such that the matrix

$$T_q = (\text{coeff}(w, q))_{w \in \mathcal{W}_q, b \in \mathcal{B}_q} \quad (10)$$

is invertible. Then it follows from  $c_w = \text{coeff}(w, \Theta) = \sum_{b \in \mathcal{B}_q} c_b \text{coeff}(w, b)$  for  $w \in \mathcal{W}_q$ , and thus  $(c_w)_{w \in \mathcal{W}_q} = T_q \cdot (c_b)_{b \in \mathcal{B}_q}$ , that the coefficients  $c_b$  in (9) can be computed as

$$(c_b)_{b \in \mathcal{B}_q} = T_q^{-1} \cdot (c_w)_{w \in \mathcal{W}_q}. \quad (11)$$

Suitable choices for such a set  $\mathcal{W}_q$  and basis  $\mathcal{B}_q$  are respectively the set of Lyndon words of grade  $q$  and the corresponding Lyndon basis [6, 11], see Tables 1 and 2.

### 3.5 Order Conditions

The following main result of this section is an easy consequence of the previous considerations.

**Theorem 3 ([9, Theorem 2.3]).** *If for  $\Phi_1, \dots, \Phi_J, \Omega \in [\mathbb{C}\langle \mathcal{A} \rangle]$  the order conditions*

$$c_w = \text{coeff}(w, e^{\Phi_J} \dots e^{\Phi_1} - e^\Omega) = 0, \quad w \in \bigcup_{q=1}^p \mathcal{W}_q \quad (12)$$



**Table 2.** Lyndon words  $\mathcal{W}_q$  of grade  $q$  and Lyndon basis  $\mathcal{B}_q$  of  $\mathfrak{g}_q$  for  $\mathcal{A} = \{A_1, \dots, A_q\}$  and  $\text{grade}(A_k) = k$ .

$q$	Lyndon words	Lyndon basis
1	$A_1$	$A_1$
2	$A_2$	$A_2$
3	$A_1 A_2, A_3$	$[A_1, A_2], A_3$
4	$A_1 A_1 A_2, A_1 A_3, A_4$	$[A_1, [A_1, A_2]], [A_1, A_3], A_4$
5	$A_1 A_1 A_1 A_2, A_1 A_1 A_3, A_1 A_2 A_2, A_1 A_4, A_2 A_3, A_5$	$[A_1, [A_1, [A_1, A_2]]], [A_1, [A_1, A_3]], [[A_1, A_2], A_2], [A_1, A_4], [A_2, A_3], A_5$

are satisfied for all Lyndon words of grade  $q \leq p$ , then

$$c_w = \text{coeff}(w, e^{\Phi_J} \dots e^{\Phi_1} - e^\Omega) = 0, \quad w \in \mathcal{A}^*, \text{ grade}(w) \leq p, \quad (13)$$

and thus  $q_{\min} \geq p + 1$  in (8).

If  $e^{\Phi_J} \dots e^{\Phi_1}$  and  $e^\Omega$  are both self-adjoint, then we may assume that  $p$  is even, and (13) holds already if the order conditions (12) are satisfied only for all Lyndon words of odd grade  $q \leq p$ .

In view of Remark 1 in Subsection 3.1 we can interpret (13) as the statement

$$e^{\Phi_J} \dots e^{\Phi_1} - e^\Omega \simeq O(\tau^{p+1}),$$

i.e.,  $e^{\Phi_J} \dots e^{\Phi_1}$  is an approximation of  $e^\Omega$  of order  $p + 1$ .

### 3.6 Example

For  $\mathcal{A} = \{A, B\}$  with  $\text{grade}(A) = \text{grade}(B) = 1$  we want to determine the parameters  $a, b, c, d \in \mathbb{R}$  such that  $S = e^{bB} e^{aA} e^{cB+d[B, A, B]} e^{aA} e^{bB}$  is a 5th order approximation of  $E = e^{A+B}$ . Since the ansatz  $S$  and the expression  $E$  are both self-adjoint in the sense of Subsection 3.3, we only have to consider Lyndon words of odd grade  $\leq 4$  (i.e., of odd length  $\leq 4$  for  $\text{grade}(A) = \text{grade}(B) = 1$ ),

$$\mathcal{W} = \mathcal{W}_1 \cup \mathcal{W}_3 = \{A, B, AAB, ABB\},$$

cf. Table 1. The order conditions (12) lead to 4 polynomial equations in 4 variables  $a, b, c, d$ , for which the following Maple code computes a unique solution corresponding to

$$S = e^{\frac{1}{6}B} e^{\frac{1}{2}A} e^{\frac{2}{3}B + \frac{1}{72}[B, [A, B]]} e^{\frac{1}{2}A} e^{\frac{1}{6}B},$$

cf. (2).

```
> Physics[Setup](noncommutativeprefix = {A, B}):
> C := Physics[Commutator]:
> X := exp(b*B)*exp(a*A)*exp(c*B+d*C(B, C(A, B)))*
      exp(a*A)*exp(b*B) - exp(A+B):
> W := [[A], [B], [A, A, B], [A, B, B]]:
> eqs := [seq(simplify(wcoeff(w, X)), w in W)];
```

$$eqs := \left[ -1 + 2a, -1 + 2b + c, -\frac{1}{6} + 2a^2b + \frac{1}{2}a^2c, -\frac{1}{6} + \frac{1}{2}ac^2 + acb + ab^2 - d \right]$$

$$> sol := solve(eqs);$$

$$sol := \left\{ a = \frac{1}{2}, b = \frac{1}{6}, c = \frac{2}{3}, d = \frac{1}{72} \right\}$$

Next we compute the leading error term  $\Theta$  of the approximation  $S$  of  $E = e^{A+B}$ , cf. (8). To this end we take  $\mathcal{W}_5$  and  $\mathcal{B}_5$  from Table 1 and compute  $T_5$ ,  $(c_w)_{w \in \mathcal{W}_5}$ , and  $(c_b)_{b \in \mathcal{B}_5}$  according to (10), (11):

```
> W5 := [[A, A, A, A, B], [A, A, A, B, B], [A, A, B, A, B],
          [A, A, B, B, B], [A, B, A, B, B], [A, B, B, B, B]];
> B5 := [C(A, C(A, C(A, C(A, B))))], C(A, C(A, C(C(A, B), B))),
          C(C(A, C(A, B)), C(A, B)), C(A, C(C(C(A, B), B), B)),
          C(C(A, B), C(C(A, B), B)), C(C(C(C(A, B), B), B), B)];
> T5 := Matrix([seq([seq(wcoeff(w, b), b in B5]), w in W5)]);
```

$$T5 := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & -2 & 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -3 & 1 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> c_w := [seq(wcoeff(w, subs(sol, X)), w in W5)];
```

$$c_w := \left[ \frac{1}{2880}, \frac{-7}{8640}, \frac{1}{480}, \frac{7}{12960}, \frac{-1}{720}, \frac{-41}{155520} \right]$$

```
> c_b := evalm(LinearAlgebra[MatrixInverse](T5) &* c_w);
```

$$c_b := \left[ \frac{1}{2880}, \frac{-7}{8640}, \frac{1}{2160}, \frac{7}{12960}, \frac{1}{4320}, \frac{-41}{155520} \right]$$

Altogether, we obtain a representation of the leading error term,

$$\begin{aligned} e^{\frac{1}{6}B} e^{\frac{1}{2}A} e^{\frac{2}{3}B + \frac{1}{72}[B, [A, B]]} e^{\frac{1}{2}A} e^{\frac{1}{6}B} - e^{A+B} &= \Theta + \dots \\ &= \frac{1}{2880}[A, [A, [A, [A, B]]]] - \frac{7}{8640}[A, [A, [[A, B], B]]] + \frac{1}{2160}[[A, [A, B]], [A, B]] \\ &\quad + \frac{7}{12960}[A, [[[A, B], B], B]] + \frac{1}{4320}[[A, B], [[A, B], B]] - \frac{41}{155520}[[[[A, B], B], B], B] + \dots \end{aligned}$$

Here, the dots represent terms of grade higher than five.

## 4 Magnus-Type Integrators

In this section we apply the theory of Section 3 with the aim of constructing Magnus-type integrators for the numerical solution of non-autonomous evolution equations

$$\partial_t u(t) = A(t)u(t), \quad t \geq t_0, \quad u(t_0) = u_0, \quad A(t) \in \mathbb{C}^{d \times d}.$$

One step  $(t_n, u_n) \mapsto (t_{n+1}, u_{n+1})$  of step-size  $\tau$  of such an integrator is given by

$$t_{n+1} = t_n + \tau, \quad u_{n+1} = \mathcal{S}(\tau, t_n)u_n, \quad (14)$$

where  $\mathcal{S}(\tau, t_n) \approx \mathcal{E}(\tau, t_n)$  approximates the exact local solution operator

$$\mathcal{E}(\tau, t_n) = e^{\Omega(\tau, t_n)} \quad (15)$$

given by the *Magnus* series  $\Omega = \Omega(\tau, t_n)$ , see [8, Section IV.7].

#### 4.1 Legendre expansions

To construct  $\mathcal{S}(\tau, t_n)$  we expand  $A(t_n + t)$  on the interval  $[t_n, t_n + \tau]$  into a series of shifted Legendre polynomials,

$$A(t_n + t) = A_1 \tilde{P}_0(t) + A_2 \tilde{P}_1(t) + A_3 \tilde{P}_2(t) + \dots, \quad t \in [0, \tau]$$

with

$$\tilde{P}_k(t) = \frac{1}{\tau} P_k\left(\frac{t}{\tau}\right), \quad P_k(x) = (-1)^k \sum_{j=0}^k \binom{k}{j} \binom{k+j}{j} (-1)^j x^j,$$

see [1, Section 3.1]. The matrix-valued coefficients  $A_k$  given by

$$A_k = (2k-1)\tau \int_0^1 P_{k-1}(x) A(t_n + \tau x) dx \quad (16)$$

depend on both  $t_n$  and  $\tau$  and satisfy

$$A_k = \mathcal{O}(\tau^k). \quad (17)$$

In terms of these coefficients the Magnus series in (15) is given by

$$\begin{aligned} \Omega = & A_1 - \frac{1}{6}[A_1, A_2] + \frac{1}{60}[A_1, [A_1, A_3]] - \frac{1}{60}[A_2, [A_1, A_2]] \\ & + \frac{1}{360}[A_1, [A_1, [A_1, A_2]]] - \frac{1}{30}[A_2, A_3] + \dots \end{aligned} \quad (18)$$

see [1, Section 3.2].

#### 4.2 Order Conditions for Magnus-Type Integrators

We consider Magnus-type integrators (14) of the form

$$\mathcal{S}(\tau, t_n) = e^{\tilde{\Phi}_J(\tau, t_n)} \dots e^{\tilde{\Phi}_1(\tau, t_n)}, \quad (19)$$

where the  $\tilde{\Phi}_j$  are linear combinations of (commutators of) approximations  $\tilde{A}_k \approx A_k$  obtained by applying a suitable quadrature formula to (16). To apply the theory of Section 3 to such integrators, we note that (19) formally corresponds to an expression

$$S = e^{\Phi_J} \dots e^{\Phi_1}$$

with Lie elements  $\Phi_j \in [\mathbb{C}\langle \mathcal{A} \rangle]$  over an alphabet  $\mathcal{A} = \{A_1, A_2, \dots\}$  with symbols  $A_k$  representing  $\tilde{A}_k \approx A_k$ , and with a grading

$$\text{grade}(A_k) = k$$

corresponding to (17), cf. Remark 1 in Subsection 3.1.

To set up order conditions according to Theorem 3 we need the Lyndon words from Table 2, and, furthermore, we have to consider  $e^\Omega$  with  $\Omega$  from (18), which is self-adjoint in the sense of Subsection 3.3. For coefficients of words  $w \in \mathcal{A}^*$  in  $e^\Omega$  (which in principle could be calculated with the algorithm of Section 2) we use the explicit formula

$$\text{coeff}(A_{d_1} \cdots A_{d_\ell}, e^\Omega) = \sum_{\substack{(k_1, \dots, k_\ell) \\ 1 \leq k_l \leq d_l}} \prod_{j=1}^{\ell} \frac{(-1)^{d_j+k_j} \binom{d_j-1}{k_j-1} \binom{d_j+k_j-2}{k_j-1}}{\sum_{i=j}^{\ell} k_i} \quad (20)$$

proven in [9, Theorem 4.1].

### 4.3 8th Order Commutator-Free Magnus-Type Integrators

In this section we construct 8th order self-adjoint commutator-free integrators involving a minimum number of exponentials. In an ansatz for such a scheme only the generators  $A_1, A_2, A_3, A_4$  have to be considered, because it can be shown that  $\text{coeff}(w, e^\Omega) = 0$  for all words  $w$  of  $\text{grade}(w) \leq 8$  containing  $A_k$  with  $k \geq 5$ , see [1, Section 3.3].<sup>9</sup> Corresponding to 22 Lyndon words of odd grade  $\leq 8$  over the alphabet  $\mathcal{A} = \{A_1, A_2, A_3, A_4\}$  there are 22 order-conditions to be considered, see Theorem 3. This implies an ansatz involving 11 exponentials and 22 parameters to be determined; an 8th order scheme of this form was derived in [1, Section 4.4]. Also with this approach we found in [9, Section 4.4] a scheme where some exponentials commute, which can thus be joined together, resulting in an 8th order scheme involving only 8 exponentials. These schemes were found by a rather brute force computation. In contrast, using the following Maple code we are able to compute the coefficients of all 8th order self-adjoint schemes with 8 exponentials in a more systematic and efficient<sup>10</sup> way.

First, we define the self-adjoint ansatz for a scheme involving 8 exponentials.

```
> Physics[Setup](noncommutativeprefix = {A}):
> S := exp(f11*A1-f12*A2+f13*A3-f14*A4)*
      exp(f21*A1-f22*A2+f23*A3-f24*A4)*
      exp(f31*A1-f32*A2+f33*A3-f34*A4)*
      exp(f41*A1-f42*A2+f43*A3-f44*A4)*
      exp(f41*A1+f42*A2+f43*A3+f44*A4)*
      exp(f31*A1+f32*A2+f33*A3+f34*A4)*
      exp(f21*A1+f22*A2+f23*A3+f24*A4)*
      exp(f11*A1+f12*A2+f13*A3+f14*A4):
```

<sup>9</sup> Of course, this follows also from (20) by a direct computation.

<sup>10</sup> Compared with the effort indicated in [1, Section 4.4].

Next we set up the 8 equations corresponding to the 8 Lyndon words of odd grade  $\leq 8$  involving only the generators  $A_1, A_2$ . The right-hand sides of these equations were computed using (20) and are hardcoded here for simplicity.

```
> W12 := [[A1], [A1, A2], [A1, A1, A1, A2], [A1, A2, A2],
           [A1, A1, A1, A1, A1, A2], [A1, A1, A1, A2, A2],
           [A1, A1, A2, A1, A2], [A1, A2, A2, A2]]:
> rhs12 := [1, -1/6, -1/40, 1/60, -1/1008, 1/420, 1/2520, -1/840]:
> vars12 := [f11, f21, f31, f41, f12, f22, f32, f42]:
> eqs12 := [seq(expand(wcoeff(w, S)), w in W12)] - rhs12:
```

We now try to solve this system of equations. After a few minutes of computing time on a standard desktop PC, Maple finds a symbolic representation (involving `RootOfs`, which are Maple representations for roots of polynomial equations) of the general solution of the system, for which we compute all possible values in numerical form. It turns out that this computation (which again takes a few minutes) has to be done with very high precision, otherwise the results do not represent reasonable solutions with small residuals if substituted into the equations.

```
> sols12 := solve(eqs12, vars12):
> Digits := 200:
> FF := seq(evalf(allvalues(sol)), sol in sols12):
```

We obtain 99 solutions altogether, 17 real solutions, and modulo complex conjugation 41 different complex solutions. Each solution determines 8 parameters of the ansatz  $S$ . There remain 8 parameters to be determined compared with 14 order conditions corresponding to the  $14 = 22 - 8$  remaining Lyndon words over  $\mathcal{A} = \{A_1, A_2, A_3, A_4\}$  of odd grade  $\leq 8$ . It is remarkable that the resulting over-determined system of equations always has a solution. To find a theoretical explanation for this fact is the topic of current investigations. Here, however, it is verified by a direct computation.

We select<sup>11</sup> one of the previously obtained 99 sets of 8 parameters, substitute it into the ansatz  $S$ , and set up 4 equations corresponding to 4 (out of 9) selected Lyndon words over  $\mathcal{A}$  of odd grade  $\leq 8$  involving  $A_3$  but not  $A_4$ . The resulting system of equation is linear and readily solved. That this solution solves also the equations corresponding to the  $5 = 9 - 4$  not selected Lyndon words will be verified below.

```
> F12 := FF[78]:
> W3 := [[A3], [A1, A1, A3], [A2, A3], [A1, A1, A1, A1, A3],
          [A1, A1, A2, A3], [A1, A1, A3, A2], [A1, A2, A1, A3],
          [A1, A3, A3], [A2, A2, A3]]:
```

<sup>11</sup> This is done for the purpose of presentation, the following considerations apply to each of the 99 parameter sets. The solution selected for this presentation leads to a particularly small local error. Note that the selected index 78 may belong to different parameter sets in different runs of the code.

```

> rhs3 := [0, 1/60, -1/30, 1/420, -1/168, 1/280, -1/840,
           1/420, -1/210]:
> vars3 := [f13, f23, f33, f43]:
> eqs3 := [seq(expand(wcoeff(w, subs(F12, S))), w in W3[2 .. 5])]
           - rhs3[2 .. 5]:
> F3 := op(solve(eqs3, vars3)):

```

Analogously as before, we substitute the 12 already obtained parameters into the ansatz  $S$ , set up 4 equations corresponding to 4 (out of 5) selected Lyndon words over  $\mathcal{A}$  of odd grade  $\leq 8$  involving  $A_4$ , and solve the resulting linear system of equations. That the obtained solution solves also the equation corresponding to the not selected Lyndon word will again be verified below.

```

> W4 := [[A1, A4], [A1, A1, A1, A4], [A1, A2, A4],
          [A1, A4, A2], [A3, A4]]:
> rhs4 := [0, -1/840, 1/210, -1/140, -1/70]:
> vars4 := [f14, f24, f34, f44]:
> eqs4 := [seq(expand(wcoeff(w, subs(F12, F3, S))),
              w in W4[1 .. 4])] - rhs4[1 .. 4]:
> F4 := op(solve(eqs4, vars4)):

```

Finally we print the calculated solution representing the 16 parameters  $f_{j,k}$  of the ansatz  $S$  and compute its residual with respect to the order conditions corresponding to all Lyndon words over  $\mathcal{A}$  of odd grade  $\leq 8$  (including those not previously selected). The tiny residual confirms that the obtained scheme indeed satisfies the order conditions for order  $p = 8$  of Theorem 3.

```

> for y in [op(F12), op(F3), op(F4)] do
  lprint(evalf(y, 50))
end do:
f11 = -1.1210783473381738227756934594506597445892745485109
f21 = 1.3210319274244662988569102191161576010502669814859
f31 = -.11488794115695215928140654449977903918312514606917
f41 = .41493436107065968320018978483428118272213271309425
f12 = 1.0089705126043564404981241135055937701303470936598
f22 = -1.1889339712738696420578749909323697235681087890036
f32 = 0.44866039420480983666929215062389499923245100101695e-1
f42 = -.13197275582656085011222031954705867101347489961070
f13 = -.78475484313672167594298542161546182121249218395766
f23 = .92477328275109744272940525314314765421496759253486
f33 = 0.24950727790821017623386132247659342458740875944374e-1
f43 = -.16496916740519678440980596377534517546121628452158
f14 = .44843133893526952911027738378026389783570981940438
f24 = -.52881775248948867348601923353730351864984279845615
f34 = -0.24298790613584639672784191664606712944260031094723e-1
f44 = .19795913373984127516833047932058800652021234941605
> W := [op(W12), op(W3), op(W4)]:

```

**Table 3.** Coefficients  $a_{j,k}$  for an 8th order commutator-free Magnus-type integrator (22).

$k = 1$	$k = 2$	$k = 3$	$k = 4$
-1.232611007291861933e+0	1.381999278877963415e-1	-3.352921035850962622e-2	6.861942424401394962e-3
1.452637092757343214e+0	-1.632549976033022450e-1	3.986114827352239259e-2	-8.211316003097062961e-3
-1.783965547974815151e-2	-8.850494961553933912e-2	-1.299159096777419811e-2	4.448254906109529464e-3
-2.982838328015747208e-2	4.530735723950198008e-1	-6.781322579940055086e-3	-1.529505464262590422e-3
-1.529505464262590422e-3	-6.781322579940055086e-3	4.530735723950198008e-1	-2.982838328015747208e-2
4.448254906109529464e-3	-1.299159096777419811e-2	-8.850494961553933912e-2	-1.783965547974815151e-2
-8.211316003097062961e-3	3.986114827352239259e-2	-1.632549976033022450e-1	1.452637092757343214e+0
6.861942424401394962e-3	-3.352921035850962622e-2	1.381999278877963415e-1	-1.232611007291861933e+0

```
> RHS := [op(rhs12), op(rhs3), op(rhs4)]:
> printf("%.5e", max(map(abs, [seq(wcoeff(w,
      subs(F12, F3, F4, S)), w in W)]-RHS))):
8.82689e-143
```

In the scheme

$$S = \prod_{j=8,\dots,1} \exp\left(\sum_{k=1}^4 f_{j,k} \mathbf{A}_k\right) \quad (21)$$

with parameters  $f_{j,k}$  calculated by the above Maple code, the  $\mathbf{A}_k$  represent Legendre expansion coefficients defined by integrals (16). To obtain an effective numerical method we have to approximate these integrals using a suitable quadrature formula. Therefore we substitute

$$\mathbf{A}_k \rightarrow (2k-1)\tau \sum_{l=1}^K w_k P_{k-1}(x_l) A(t_n + \tau x_l)$$

in (21) with Gaussian nodes and weights of order eight,

$$(x_k) = \left( \frac{1}{2} - \sqrt{\frac{15+2\sqrt{30}}{140}}, \frac{1}{2} - \sqrt{\frac{15-2\sqrt{30}}{140}}, \frac{1}{2} + \sqrt{\frac{15-2\sqrt{30}}{140}}, \frac{1}{2} + \sqrt{\frac{15+2\sqrt{30}}{140}} \right),$$

$$(w_k) = \left( \frac{1}{4} - \frac{\sqrt{30}}{72}, \frac{1}{4} + \frac{\sqrt{30}}{72}, \frac{1}{4} + \frac{\sqrt{30}}{72}, \frac{1}{4} - \frac{\sqrt{30}}{72} \right),$$

which corresponds to an application of Gaussian quadrature to (16). For the set of parameters  $\{f_{j,k}\}$  displayed in the above Maple code we obtain the integrator (cf. (14), (19))

$$\mathcal{S}(t_n, \tau) = \prod_{j=8,\dots,1} \exp\left(\tau \sum_{k=1}^4 a_{j,k} A(t_n + \tau x_k)\right) \quad (22)$$

with coefficients  $a_{j,k}$  given in Table 3. For these coefficients the positivity condition

$$\operatorname{Re} f_{j,1} = \operatorname{Re} \sum_{k=1}^4 a_{j,k} > 0, \quad j = 1, \dots, 8, \quad (23)$$

**Table 4.** Real (top) and imaginary (bottom) parts of coefficients  $a_{j,k}$  for an 8th order commutator-free Magnus-type integrator (22) satisfying the positivity condition (23).

$k = 1$	$k = 2$	$k = 3$	$k = 4$
5.162172083124911076e-2	-5.787809823308952456e-3	1.404202563971892685e-3	-2.873779919999358082e-4
1.129000600487386325e-1	-1.811008163470541820e-2	8.982553129811831365e-3	-2.544930699554437791e-3
2.631601314221973826e-2	1.983998701294184106e-1	-4.965939955061425298e-2	1.197843408520720342e-2
-1.592059248033346570e-2	1.424220211513735403e-1	4.842122146532602005e-2	-1.013590436679991693e-2
-1.013590436679991693e-2	4.842122146532602005e-2	1.424220211513735403e-1	-1.592059248033346570e-2
1.197843408520720342e-2	-4.965939955061425298e-2	1.983998701294184106e-1	2.631601314221973826e-2
-2.544930699554437791e-3	8.982553129811831365e-3	-1.811008163470541820e-2	1.129000600487386325e-1
-2.873779919999358082e-4	1.404202563971892685e-3	-5.787809823308952456e-3	5.162172083124911076e-2
-1.187198036084005914e-1	1.331082409655082917e-2	-3.229389682031679030e-3	6.609128526175740449e-4
1.359790143178213473e-1	3.226637801235380303e-3	-5.647440118497178834e-3	1.831962429052182520e-3
-1.952925932474600076e-2	4.339859420803126316e-2	4.884840043796339250e-3	-1.849278537972746835e-3
3.513884130112852023e-3	-7.185755041597012718e-2	1.591348406688517315e-2	-1.887432258484616938e-3
-1.887432258484616938e-3	1.591348406688517315e-2	-7.185755041597012718e-2	3.513884130112852023e-3
-1.849278537972746835e-3	4.884840043796339250e-3	4.339859420803126316e-2	-1.952925932474600076e-2
1.831962429052182520e-3	-5.647440118497178834e-3	3.226637801235380303e-3	1.359790143178213473e-1
6.609128526175740449e-4	-3.229389682031679030e-3	1.331082409655082917e-2	-1.187198036084005914e-1

is not satisfied, in agreement with the fact that this cannot be the case for real coefficients, see [10]. For some applications, however, it is essential for stability reasons that this condition is satisfied. This suggests to consider schemes with complex coefficients, see [4]. As was mentioned above, of the 99 parameter sets  $\{f_{j,k}\}$  which can be computed by the above Maple code,  $82 = 41 \times 2$  involve complex numbers. One of these parameter sets leads to the scheme (22) with coefficients given in Table 4, for which (23) is satisfied.

**Acknowledgements.** This work was supported in part by the Austrian Science Fund (FWF) under grant P30819-N32 and the Vienna Science and Technology Fund (WWTF) under grant MA14-002.

## References

1. Alverman, A., Fehske, H.: High-order commutator-free exponential time-propagation of driven quantum systems. *J. Comput. Phys.* **230**, 5930–5956 (2011)
2. Auzinger, W., Herfort, W.: Local error structures and order conditions in terms of Lie elements for exponential splitting schemes. *Opuscula Math.* **34**, 243–255 (2014)
3. Auzinger, W., Herfort, W., Hofstätter, H., Koch, O.: Setup of order conditions for splitting methods. In: Gerdt, V.P., Koepf, W., Seiler, W.M., Vorozhtsov, E.V. (eds.) *Computer Algebra in Scientific Computing*. pp. 30–42. Springer International Publishing, Cham (2016)
4. Blanes, S., Casas, F., Thalhammer, M.: High-order commutator-free quasi-Magnus exponential integrators for nonautonomous linear evolution equations. *Comput. Phys. Commun.* **220**, 243–262 (2017)
5. Chin, S.: Symplectic integrators from composite operator factorizations. *Physics Letters A* **226**(6), 344–348 (1997)
6. Duval, J.: Génération d’une section des classes de conjugaison et arbre des mots de Lyndon de longueur bornée. *Theoret. Comput. Sci.* **60**, 255–283 (1988)
7. Expocon.mpl, <https://github.com/HaraldHofstaetter/Expocon.mpl>



8. Hairer, E., Lubich, C., Wanner, G.: Geometric Numerical Integration. Springer-Verlag, Berlin–Heidelberg–New York, 2nd edn. (2006)
9. Hofstätter, H.: Order conditions for exponential integrators (Feb 2019), submitted (for a preprint see arXiv:1902.11256).
10. Hofstätter, H., Koch, O.: Non-satisfiability of a positivity condition for commutator-free exponential integrators of order higher than four. *Numer. Math.* **141**(3), 681–691 (2019)
11. Lothaire, M.: Combinatorics on Words. Encyclopedia of Mathematics and its Applications, Cambridge University Press, Cambridge, U.K. (1997)
12. Munthe-Kaas, H., Owren, B.: Computations in a free Lie algebra. *Phil. Trans. R. Soc. Lond. A* **357**(1754), 957–981 (1999)
13. Suzuki, M.: New scheme of hybrid exponential product formulas with applications to quantum Monte-Carlo simulations. In: *Computer Simulation Studies in Condensed-Matter Physics VIII*. Springer-Verlag, Berlin, Heidelberg (1995)